
1 Introdução
A linguagem Untyped L1 é uma linguagem minimalista com propósito di-
dático. Ela possui as seguintes características:

• booleanos representados como inteiros (0 representa falso, qualquer
outro número representa verdadeiro)

• operações binárias e unárias são escritas de forma prefixada (como em
LISP/Scheme). Por exemplo, 4+5*6 é escrito (+ 4 (* 5 6))

• funções anônimas (expressões lambda)

• funções recursivas (letrec)

• linguagem funcional pura (sem efeitos colaterais e sem rotinas de en-
trada e saída de dados)

A seguir são apresentadas de forma concisa a sintaxe concreta, sintaxe
abstrata e semântica formal da linguagem. Note que, apesar de não pos-
suir nenhuma estrutura de dados pré-definida (como pares ou listas), essas
podem ser codificadas através de termos lambda, como será apresentado
através de um exemplo.

2 Sintaxe Concreta (BNF)
<uop> ::= not
<bop> ::= and | or | + | - | * | > | < | ==
<num> ::= [-]?[0..9][0..9]*
<id> ::= [a..z][a..z,0..9,_]*
<exp> ::=
| <id> [variável]
| <num> [inteiro]
| ( <exp> ) [parênteses]
| \ <id> . <exp> [função de um parâmetro]
| <exp> <exp> [aplicação de função a argumento]
| <uop> <exp> [operador unário]
| <bop> <exp> <exp> [operador binário]
| if <exp> then <exp> else <exp> [condicional]
| let <id> = <exp> in <exp> [definição local]
| letrec <id> = \ <id> . <exp> in <exp> [definição de função recursiva]

3 Sintaxe abstrata

Expressões:

e ::= n
| uop e
| e1 bop e2
| if e1 then e2 else e3
| x
| e1 e2
| λ x ⇒ e
| let x = e1 in e2
| let rec f= (λ x ⇒ e1) in e2

onde

n ∈ {. . . ,−2,−1, 0, 1, 2, . . . }
x ∈ Ident (conjunto de identificadores)
uop ∈ {¬}
bop ∈ {+,−, ∗ >,<,=,∨,∧}

Valores:
v ::= n | λ x ⇒ e

4 Semântica

[[n]] = [[n1]] op [[n2]] op ∈ {+,−, ∗}
n1 op n2 −→ n

(opArith)

[[n1]] r [[n2]] r ∈ {∧,∨,=, <,>}
n1 r n2 −→ 1

(opRel1)

[[n1]] ̸ r [[n2]] r ∈ {∧,∨,=, <,>}
n1 r n2 −→ 0

(opRel2)

not 0 −→ 1
(opNot1)

n ̸= 0

not n −→ 0
(opNot2)

e1 −→ e′1

e1 bop e2 −→ e′1 bop e2
(op1)

e2 −→ e′2

v bop e2 −→ v bop e′2
(op2)

e −→ e′

not e −→ not e′
(op3)

n ̸= 0

if n then e2 else e3 −→ e2
(if1)

if 0 then e2 else e3 −→ e3 (if2)

e1 −→ e′1

if e1 then e2 else e3 −→ if e′1 then e2 else e3
(if3)

(λ x ⇒ e) v −→ {v/x}e (β)

e2 −→ e′2

v e2 −→ v e′2
(app1)

e1 −→ e′1

e1 e2 −→ e′1 e2
(app2)

let x = v in e2 −→ {v/x}e2 (let1)

e1 −→ e′1

let x = e1 in e2 −→ let x = e′1 in e2
(let2)

let rec f= (λ y ⇒ e1) in e2
−→

{(λ y ⇒ let rec f= (λ y ⇒ e1) in e1)/f} e2

(letrec)

5 Exemplo de código
O código a seguir apresenta diversos elementos interessantes:

• codificação de pares e listas como funções

• implementação de um algoritmo de ordenamento de listas

• implementação da função fatorial

• implementação da função map

Através das funçõe definidas, é feita a seguinte chamada (em sintaxe de
LISP)

(map fatorial (sort (list 7 4 5 3)))
==>
(list 6 24 120 5040)

Note que os números da lista resultante vêm precedidos de 0’s na represen-
tação em árvore (devido à codificação de listas como funções).

Código de exemplo:

let pair = \a. \b. \c. if c then a else b in
let fst = \p. p 1 in
let snd = \p. p 0 in

let empty = pair 1 1 in
let cons = \h. \t. pair 0 (pair h t) in
let head = \l. fst (snd l) in
let tail = \l. snd (snd l) in
let isEmpty = \l. fst l in

letrec
insert =

\v. \l. if (isEmpty l)
then (cons v empty)
else if (> (head l) v)

then (cons v l)
else (cons (head l) (insert v (tail l)))

in

letrec
sort = \l. if isEmpty l

then l
else (insert (head l) (sort (tail l)))

in

letrec
fatorial = \x. if (> x 1)

then (* x (fatorial (+ x (-1))))
else 1

in

letrec
map = \f. \l. if isEmpty l

then l
else (cons (f (head l)) (map f (tail l)))

in

map fatorial (sort (cons 7 (cons 4 (cons 5 (cons 3 empty)))))

